

Verified and Optimized Implementation of Orthologic Proof Search

Simon Guilloud¹ and Clément Pit-Claudel²

EPFL

School of Computer and Communication Sciences
Station 14, CH-1015 Lausanne, Switzerland

{firstname.lastname}@epfl.ch

Abstract. We report on the development of an optimized and verified decision procedure for orthologic equalities and inequalities. This decision procedure is quadratic-time and is used as a sound, efficient and predictable approximation to classical propositional logic in automated reasoning tools. We start by formalizing, in the Coq proof assistant, a proof system in sequent-calculus style for orthologic. We then prove its soundness and completeness with respect to the algebraic variety of ortholattices, and we formalize a cut-elimination theorem (in doing so, we discover and fix a missing case in a previously published proof).

We then implement and verify a complete proof search procedure for orthologic. A naive implementation is exponential, and to obtain an optimal quadratic runtime, we optimize the implementation by memoizing its results and simulating reference equality testing. We leverage the resulting correctness theorem to implement a reflective Coq tactic. We present benchmarks showing the procedure, under various optimizations, matches its theoretical complexity.

1 Introduction

Specialized, reliable and efficient building blocks are indispensable in scaling automated reasoning software. Program verifiers, SMT solvers, proof assistants and automated theorem provers use them to tackle the various theories and subproblems that compounds a logical statement. As they combine, so do the possibility of an implementation error. To ensure that they can be used as trusted components in program verification pipelines, it is very desirable to verify these verification algorithms themselves.

One fragment of particular interest for the vast majority of automated reasoning software is propositional logic, i.e. Boolean algebra. Despite significant progress in heuristics, solving satisfiability or validity of propositional formulas remains a major challenge to scalability of decision procedures. An alternative and complementary approach is orthologic, a non-distributive generalization of classical propositional logic, which admits polynomial-time ($\mathcal{O}(n^2)$) validity checking and normalization algorithms [2,8,11]. Orthologic offers a trade-off: it

sacrifices completeness (with respect to classical semantics) in exchange for guaranteed efficiency and predictability. Normalization in particular allows to compute a unique normal form (with respect to the laws of orthologic) of guaranteed minimal size. This helps with caching, solves a variety of intuitive formula equivalences, and allows simplifying formulas before solving them with more complete but possibly inefficient procedures. Recent work has demonstrated the practical utility of orthologic-based reasoning, being an important component of verification tools such as the Stainless program verifier [8] and the Lisa proof assistant [9]. Table 1 presents the laws of orthologic.

V1: $x \vee y = y \vee x$	V1': $x \wedge y = y \wedge x$
V2: $x \vee (y \vee z) = (x \vee y) \vee z$	V2': $x \wedge (y \wedge z) = (x \wedge y) \wedge z$
V3: $x \vee x = x$	V3': $x \wedge x = x$
V4: $x \vee 1 = 1$	V4': $x \wedge 0 = 0$
V5: $x \vee 0 = x$	V5': $x \wedge 1 = x$
V6: $\neg\neg x = x$	
V7: $x \vee \neg x = 1$	V7': $x \wedge \neg x = 0$
V8: $\neg(x \vee y) = \neg x \wedge \neg y$	V8': $\neg(x \wedge y) = \neg x \vee \neg y$
V9: $x \vee (x \wedge y) = x$	V9': $x \wedge (x \vee y) = x$

Table 1. Laws of ortholattices, algebraic varieties with signature $(S, \wedge, \vee, 0, 1, \neg)$.

In this work, we present the first formalization and verification, using the Coq proof assistant, of an efficient decision procedure for the validity and equivalence problems for propositional formulas in orthologic. This algorithm [11] is based on proof search in a proof system for orthologic that is a restriction of the classical sequent calculus where a sequent can never contain more than two distinct formulas at a time. Despite this restriction, the proof system admits cut elimination, which we prove. The cut elimination property implies a subformula property, which is key to the completeness of the proof search procedure.

A naive implementation of the decision procedure has exponential complexity. To obtain a polynomial version, we leverage memoization, i.e. storing in a table the intermediate results of the recursive calls. Moreover, using structural equality to check if a key is in the memoization map costs an additional linear runtime factor. To obtain an optimal quadratic version, we modify the algorithm to use reference equality. As our algorithm is purely functional, this means we have to extend our formulas abstract syntax trees to assign to each node a unique identifier (or pointer) that can then be used in the memoization map. We formally prove the correctness of these constructions.

We implement and verify four versions of the algorithm: without optimization ($\tilde{O}(2^n)$), with memoization using lists ($\tilde{O}(n^5)$), using AVL maps from the Coq standard library ($\tilde{O}(n^3)$), and using AVL maps and simulated reference equality ($\tilde{O}(n^2)$). Memoization and reference equality are generic, and important in many tools and algorithms. Our approach to verifying these optimizations is not

specific to orthologic proof search and we expect it to extend to any recursive algorithms over algebraic data types.

Using the technique of proofs by reflection, we then obtain an efficient, executable proof tactic that decides equality modulo orthologic rule and which is applicable to any ortholattice, in particular the type `bool` of boolean values. This tactic is able to solve automatically, for example,

```
false = (a ∩ b) ∪ ¬(a) ∪ ¬(b)
```

. We present benchmarks attesting that each version of the proof search procedure (with memoization using a list-based map, with memoization using AVL trees, and with both memoization and reference equality) meets the theoretical complexity.

1.1 Contributions

In [section 2](#), we first formalize in the Coq proof assistant the following elements: a type class of ortholattices, a proof system for orthologic (in the style of sequent calculus), its soundness and completeness with respects to truth in every ortholattice, and other useful properties.

In [section 3](#), we then formalize the cut elimination theorem (without non-logical axioms) of orthologic, and in the process discover a missing edge case in the original proof from [\[11\]](#). The proof is syntactic and proceeds with a large case analysis on the structure of a correct orthologic proof, with parallel induction on multiple measures of the input. We ensure the correctness of the proof using dependent types.

In [section 4](#), we implement the proof-search based decision procedure for orthologic from [\[11\]](#). From this and the soundness theorem of orthologic, we use reflection to obtain a Coq proof tactic which can solve any equality or inequality valid in orthologic, for any ortholattice. A special case of ortholattice of interest is the type `bool` of truth values. The tactic could in particular efficiently certify the reduction of a formula to its *OL normal form*, a strengthening of the negation normal proposed in [\[8\]](#).

In [section 5](#), we describe our implementation and verification of the memoized version of the algorithm. As the algorithm is fully functional, the memoization map is implemented as a state monad.

In [section 6](#), we discuss reference equality. This optimization is significantly more complex. It requires redefining trees representing orthologic terms to include pointers and proving that the transformation from pointer-free terms to terms with pointers is correct, which was surprisingly difficult.

In [section 7](#), we present a series of benchmark which are supposed to hit the asymptotic worst case of every implementation, using different reduction strategies, and find that the resulting curves match the theoretical complexity.

1.2 Related Work

The word problem for ortholattices (deciding if given two terms, one is always \leq than the other) was first solved by [\[2\]](#), extended in [\[8\]](#) to obtain normal forms

of terms. A similar proof system for orthologic, with the property that sequents never contain more than one formula, was already described by [19]. Different proof systems have also been considered [17,16,15].

The authors of [3] formalize cut elimination for intuitionistic logic and [6] cut elimination for the logic of bunched implications in Coq, but their approach is semantic rather than syntactic. [22], still using Coq, show Cut elimination for a class of coalgebraic logics, including a variety of modal logics. They propose both a syntactic and a semantic proof. In Isabelle, [24] formalizes a particular, strongly normalizing version of cut elimination for classical first order logic.

Whitman’s algorithm, a decision procedure for inequalities holding in every lattice (without negation), has been formalized in Coq by [13], also using reflection.

Ortholattices have been defined in Mizar [23], but little about them proven. As far as we know, no result nor even definition in ortholattice theory have been formalized in any other proof assistant.

Verified memoization has been studied in [25]. The authors propose a framework for automatic verified memoization of programs in Isabelle/HOL. In Coq, [1] studies the related topic of hash-consing, which can be seen as memoization of constructors. In their approach, they entirely replace nodes of an ADT by identifier, so that the recursive structure only exist in the hash-consing map, while we add the identifier to the tree.

2 Formalizing Ortholattices and Orthologic

We first formalize the algebraic class of ortholattices as a typeclass (Listing 1.1). Ortholattices are of course *lattices*, and in particular a partial order with \leq given by

$$x \leq y \iff x \wedge y = x$$

or equivalently

$$x \leq y \iff x \vee y = y.$$

Ortholattices can be axiomatized equivalently as an algebraic variety, such as in Table 1, or as a partial order. The partial order definition is closer in spirit to the proof system that we will define for Orthologic. We proved the properties of Table 1 from the definition above. To minimize proof burden when implementing the type class, we removed the axioms about 1 and 0. 1 and 0 are then defined, rather than axiomatized.

We then implement with Ltac *Whitman’s algorithm* [4], a simple decision procedure for lattices. This helps us to quickly show a number of useful lemmas about ortholattices.

```
Ltac Whitman := match goal with
| [ |- _ == _ ] => rewrite equiv_leq; split; Whitman
| [ |- ?x <= ?x ] => apply P1
| [ |- _ <= _ ∩ _ ] => apply P6; Whitman
```

```

Class Ortholattice := {
  A : Set;

  leq : relation A where "x <= y" := (leq x y);
  meet : A -> A -> A where "x ∩ y" := (meet x y);
  join : A -> A -> A where "x ∪ y" := (join x y);
  neg : A -> A where "¬ x" := (neg x);

  equiv: relation A where "x == y" := (equiv x y);
  equiv_leq : forall x y, (x == y) <-> ( (x <= y) /\ (y <= x));

  P1 : forall x, x <= x;
  P2 : forall x y z, x <= y -> y <= z -> x <= z;
  P4 : forall x y, (x ∩ y) <= x;
  P5 : forall x y, (x ∩ y) <= y;
  P6 : forall x y z, (x <= y) -> (x <= z) -> x <= (y ∩ z);
  P7 : forall x, x <= ¬( ¬( x));
  P8 : forall x y, x <= y -> ¬( y) <= ¬( x);
  P9 : forall x y, (x ∩ ¬( x)) <= y;

  P4' : forall x y, x <= (x ∪ y);
  P5' : forall x y, y <= (x ∪ y);
  P6' : forall x y z, (x <= z) -> (y <= z) -> (x ∪ y) <= z;
  P7' : forall x, ¬( ¬( x)) <= x;
  P9' : forall x y, x <= ( y ∪ ¬( y));
}.

```

Listing 1.1. Definition of an Ortholattice

```

| [ |- _ ∪ _ <= _ ] => apply P6'; Whitman
| [ |- _ ∩ _ <= _ ] =>
  try (apply glb1; Whitman; eauto; fail);
  try (apply glb2; Whitman; eauto; fail)
| [ |- _ <= _ ∪ _ ] =>
  try (apply lub1; Whitman; eauto; fail);
  try (apply lub2; Whitman; eauto; fail)
| [ |- ¬ _ <= ¬ _ ] => apply P8; Whitman
| [ |- _ <= _ ] => try (eauto;fail)
end.
Lemma example {OL: Ortholattice} a b : (a ∩ b) <= (a ∪ b).
Proof. Whitman. Qed.

```

This simple tactic is subsumed by the tactic for ortholattices we will obtain with reflection. We then show that `equiv` is a congruence relation for `<=`, `∩`, `∪` and `¬`. This makes ortholattices *setoids*, and enables the use of *generalized rewriting* [21].

Orthologic

We define in the standard way the type of ortholattice terms, and the evaluation of a term in an arbitrary ortholattice:

```

Inductive Term : Set :=
  | Var : positive -> Term
  | Meet : Term -> Term -> Term
  | Join : Term -> Term -> Term
  | Not : Term -> Term.

Definition Zero := Meet (Var 0) (Not (Var 0)).
Definition One := Join (Var 0) (Not (Var 0)).

Fixpoint eval {OL: Ortholattice}
  (t: Term) (f: nat -> A) : A :=
  match t with
  | Var n => f n
  | Meet t1 t2 => (eval t1 f) ∩ (eval t2 f)
  | Join t1 t2 => (eval t1 f) ∪ (eval t2 f)
  | Not t1 => ¬ (eval t1 f)
  end.

```

and then implement the OL proof system of [11], formulated as a sequent calculus. One can think of this proof system as Gentzen's sequent calculus for classical logic [7] restricted to ensure that at any given point in a proof, a sequent never has more than two formulas on both sides combined.

We represent sequents as (ordered) pairs of annotated formulas:

```

Inductive AnTerm : Set :=
  | N : AnTerm
  | L : Term -> AnTerm
  | R : Term -> AnTerm.

Definition Sequent (l r : AnTerm) := (l, r).

```

Where N represents no formula, L a formula on the left and R a formula on the right. For example, $(L\phi, R\psi)$ stands for $\phi \vdash \psi$ in more conventional notation.

We implemented the proof system using dependent inductive types, so that the correctness of a proof is guaranteed by construction, and no additional proof-checking function is required (Listing 1.2). The rules in mathematical notations are given in [11], but for example the `RightAnd` rule corresponds to

```

Inductive OLProof : AnTerm*AnTerm -> Set :=
| Hyp: forall {p}, OLProof (L (Var p), R (Var p))
| Weaken: forall {g} {d},
  OLProof (g, N) -> OLProof (g, d)
| Contract: forall {g},
  OLProof (g, g) -> OLProof (g, N)
| LeftAnd1: forall {a} {b} {d},
  OLProof (L a, d) -> OLProof (L (Meet a b), d)
| LeftAnd2: forall {a} {b} {d},
  OLProof (L b, d) -> OLProof (L (Meet a b), d)
| LeftOr: forall {a} {b} {d},
  OLProof (L a, d) -> OLProof (L b, d) -> OLProof (L (Join a b), d)
| LeftNot: forall {a} {d},
  OLProof (R a, d) -> OLProof (L (Not a), d)
| RightAnd: forall {a} {b} {g},
  OLProof (g, R a) -> OLProof (g, R b) -> OLProof (g, R (Meet a b))
| RightOr1: forall {a} {b} {g},
  OLProof (g, R a) -> OLProof (g, R (Join a b))
| RightOr2: forall {a} {b} {g},
  OLProof (g, R b) -> OLProof (g, R (Join a b))
| RightNot: forall {a} {g},
  OLProof (g, L a) -> OLProof (g, R (Not a))
| Swap: forall {g} {d},
  OLProof (g, d) -> OLProof (d, g)
| Cut: forall {g} {b} {d},
  OLProof (g, R b) -> OLProof (L b, d) -> OLProof (gamma, d).

```

Listing 1.2. Proof system for orthologic, with explicit Swap and Contract rules.

$$\frac{\Gamma, \phi^R \quad \Gamma, \psi^R}{\Gamma, (\phi \wedge \psi)^R} \text{RightAnd}$$

In [11], sequents are formally considered as sets. We have defined them in Coq using ordered pairs and hence needs to define two additional rules, simulating the set-like nature of sequents: the Swap and Contract rules.

Soundness and Completeness

Soundness of the proof systems states that if a sequent is provable, then the corresponding inequality must be true in every ortholattice (semantic truth). Completeness is the converse. We define the `Leq` and `AnLeq` relation denoting semantic truth for terms and annotated terms:

```

Definition term_leq (t1 t2: Term) : Prop := forall (OL: Ortholattice),

```

```
forall f: positive -> A, eval t1 f <= eval t2 f.
Definition anterml_eq (l r : AnTerm) : Prop :=
... (* generalization to annotated terms *)
```

Note that `AnLeq` is symmetric, as it would be in a presentation of classical sequent calculus with side annotations; the usual asymmetry is recovered in the specific case where one of the argument is `L`-annotated and the other is `R`-annotated. We then state and prove soundness:

```
Theorem Soundness s (p: OLProof s):
  AnLeq (fst s) (snd s).
```

Listing 1.3. The soundness theorem for orthologic

The proof proceeds by induction on the OL proof `p`. Every case follows easily from the definitions, the properties of ortholattices and some simple intermediate lemmas.

```
Theorem Completeness l r:
  Leq l r -> hasProof l r.
```

This proof follows from the fact that the type `Term` is itself an ortholattice, when the less than or equal relation is interpreted as `hasProof`. The proof of all the ortholattice properties follow straightforwardly from the proof steps.

```
#[refine] Instance TermOL : Ortholattice := {
  A := Term;
  leq := hasProof;
  meet := Meet;
  join := Join;
  neg := Not;
}. Proof ... Qed.
```

3 Cut elimination for Orthologic

The key property of the OL proof system is that it admits cut elimination: Any provable sequent can be proven without using the cut rule. Cut elimination has important theoretical and practical consequences. Since the `cut` step is the only one in which a term can appear in the premise but not the conclusion, cut elimination implies the subformula property: if a sequent has a proof, then it has a proof where only subterms of the conclusion appear. Hence, orthologic admits terminating proof search.

```
Theorem cut_elimination s (proof : OLProof s):
  {p: OLProof s | is_cut_free p}.
```

This theorem is significantly more complicated to prove than the results above. The paper proof starts with “Consider the topmost instance of the cut

rule in the proof". This corresponds to doing a double induction, first on the number of cut appearing in a proof (`fuelCut`), then on the size of the proof (`fuelSize`). The proof takes as arguments proofs that the given fuel is larger than the metrics it represents, and every induction step need to justify that the arguments are decreasing. This reduces the theorem to eliminating a single Cut step from an otherwise cut free proof.

Graphically, we have the following situation:

$$\frac{\frac{\mathcal{A}}{\Gamma, b^R} \quad \frac{\mathcal{B}}{b^L, \Delta}}{\Gamma, \Delta} \text{Cut}$$

and need to obtain the conclusion Γ, Δ with a cut free proof. The proof again works by double induction: first on the size of the cut formula b , then on the total size of \mathcal{A} and \mathcal{B} :

```
Lemma inner_cut_elim : forall
  (fuelB: nat)
  (b: Term) (good_fuelB: fuelB >= termSize b)
  (fuelSize: nat)
  (gamma: AnTerm) (delta: AnTerm)
  (A: OLProof (gamma, R b)) (p1: isCutFree A)
  (B: OLProof (L b, delta)) (p2: isCutFree B)
  (good_fuelSize: fuelSize >= (Size A + Size B)),
  {p: OLProof (gamma, delta) | isCutFree p}.
```

We then proceed by case analysis on \mathcal{A} and \mathcal{B} , and give a specific transformation in each case. For example,

$$\frac{\frac{\frac{\mathcal{A}'}{\Gamma, \alpha^R}}{\Gamma, (\alpha \vee \beta)^R} \text{ RightOr} \quad \frac{\frac{\mathcal{B}'}{\alpha^L, \Delta} \quad \frac{\mathcal{B}''}{\beta^L, \Delta}}{(\alpha \vee \beta)^L, \Delta} \text{ LeftOr}}{\Gamma, \Delta} \text{ Cut} \quad \hookrightarrow \quad \frac{\frac{\mathcal{A}'}{\Gamma, \alpha^R} \quad \frac{\mathcal{B}'}{\alpha^L, \Delta}}{\Gamma, \Delta} \text{ Cut}$$

Again, every single recursive call has to provide the required proof of decreasing measures corresponding to the fuel properties. We implemented one tactic for each induction principle that applies it correctly and automatically show the necessary fuel and cut-free property on the arguments.

The main difficulty of the proof is that the case analysis is huge: The Swap step essentially duplicates every other proof step by allowing them to act on the first or second formula, which implies we have to analyse each of \mathcal{A} and \mathcal{B} on 23 cases each, for a total of more than 500 cases. In practice, we can first do the analysis on \mathcal{A} , and for some cases the proof is independent of the structure of \mathcal{B} . However, there is no easy way to undo the case analysis on \mathcal{A} when conversely

the cases of pattern matching on \mathcal{B} have a proof independent of the structure of \mathcal{A} .

Then, some combinations of cases are impossible. For example, it is not possible for \mathcal{A} to conclude with a `RightAnd` and \mathcal{B} with `LeftOr1`, as the cut formula b would then need to be both a conjunction and a disjunction. Thanks to using dependent types to define the proof system, those cases are automatically eliminated. “Suppose \mathcal{A} concludes with a `RightOr` step. Then b must be of the form $\alpha \vee \beta$.”, but finding which combination of existing tactics simulate this was difficult. Trial and error yielded a custom tactic based on `dependent inversion` (which, unlike `dependent destruction`, does not assume additional axioms) and adequate generalization of hypothesis.

The paper proof can afford a lot of reasoning by symmetry. There are particular combined symmetries between left and right rules, meet and join, swapped and non-swapped cases, \mathcal{A} and \mathcal{B} , etc, which can easily be informally treated in a paper proof but need to be formally addressed in Coq. In the end, the formal proof contains around 200 cases.

Among these cases, we caught one in particular that was not properly considered in the paper proof of [11]. Indeed, the example reduction above would fail in the presence of an implicit contraction due to set semantics. Written with an explicit contraction, this is:

$$\frac{\frac{\frac{\mathcal{A}'}{\Gamma, \alpha^R}}{\Gamma, (\alpha \vee \beta)^R} \text{ RightOr} \quad \frac{\frac{\frac{\mathcal{B}'}{\alpha^L, (\alpha \vee \beta)^L} \quad \frac{\mathcal{B}''}{\beta^L, (\alpha \vee \beta)^L}}{(\alpha \vee \beta)^L, (\alpha \vee \beta)^L} \text{ LeftOr}}{(\alpha \vee \beta)^L, N} \text{ Contract}}{\Gamma, N} \text{ Cut}$$

And the transformation above would yield $\Gamma, (\alpha \vee \beta)^L$ instead of Γ, N . In this case, a correct transformation is as given in

$$\frac{\frac{\frac{\mathcal{A}'}{\Gamma, \alpha^R}}{\Gamma, (\alpha \vee \beta)^R} \text{ RightOr} \quad \frac{\frac{\mathcal{B}'}{(\alpha \vee \beta)^L, \alpha^L}}{\alpha^L, \Gamma} \text{ Cut}}{\frac{\frac{\Gamma, \Gamma}{\Gamma, N} \text{ Contract}}{\Gamma, N} \text{ Cut}}$$

The topmost cut is justified by induction because the proof is smaller than the original one. The second however isn't: After recursive elimination from the cut above, we can't guarantee that the new proof is smaller. Instead, this cut needs to be justified by induction on the size of the cut formula.

4 Decision Procedure for Orthologic

The *word problem* for ortholattices consists in deciding, for arbitrary terms over (\wedge, \vee, \neg) s and t , if $s = t$ in all ortholattices. Cases of particular interest involve

deciding if $s = 1$ (validity) or $s = 0$ (unsatisfiability). Since $s = t \iff s \leq t \ \& \ t \leq s$, and vice-versa $s \leq t \iff s = s \wedge t$, we can equivalently see the word problem as deciding inequality between arbitrary words. By soundness and completeness of the orthologic proof system, this is in turn equivalent to deciding if the sequent s^L, t^R has a proof.

This is decidable with a recursive backward proof search procedure, as in [11]. Given a sequent Γ, Δ , try to apply all the rules from Listing 1.2 which can conclude with Γ, Δ , and recursively solve the premise. Since orthologic admits cut elimination, we never need to consider the **Cut** step, and **Swap** never needs to be applied twice in a row. Then, all other steps reduce the size of the formula when applied backward, and hence the procedure is terminating.

In practice, this procedure can be optimized by observing that some proof steps, when applied backward, are not merely sufficient conditions but also necessary. For example, consider a sequent of the form

$$\Gamma, (\phi_1 \wedge \phi_2)^L.$$

It is a theorem of lattices that $s \leq t_1 \wedge t_2 \iff s \leq t_1 \ \& \ s \leq t_2$, corresponding to the **RightAnd** step. Hence, for $\Gamma, (s \wedge t)^R$ to have a proof, it is not only sufficient but also necessary that both Γ, s^R and Γ, t^R have a proof. It follows that if **RightAnd** is applicable, we do not need to try other steps. Listing 1.4 shows an implementation of this decision procedure, the correctness of which is stated as:

```
Theorem decideOL_base_correct :
  forall n g d,
    (decideOL_n g d) = true ->
      squash (OLProof (g, d)).
```

The proof proceeds by induction on n , and then does a large case analysis on the sequent (g, d) . Each of g and d can be of the form N, Lt or Rt , and t can be either a variable, a meet, a join or a negation, for a total of $(1+4+4)^2 = 81$ cases. Using some automation, many cases can be solved at once, and we're essentially left with constructing a correct proof for each of the ways the algorithm can return **true**.

Reflection

Suppose we have a goal $s \leq t$ or $s = t$, where s and t are expressions in an arbitrary ortholattice, for example:

```
Lemma example: forall a b : bool,
  a && negb a = negb b && (a && b).
```

We want to solve this problem by executing the above algorithm, which we proved correct. First, we split $s = t$ in two independent inequalities. Then, we *reflect* the problem into the language of orthologic expressions. To do so, we find two terms s' and t' as well as a (finite) function $f: \text{nat} \rightarrow \text{bool}$ such that $\text{eval } s' \ f$ is convertible to s and $\text{eval } t' \ f$ is convertible to t . Our goal is then convertible to:

```

Fixpoint decideOL_base (fuel: nat) (g d: AnTerm) : bool :=
  match fuel with
  | 0 => false
  | S n =>
    match (g, d) with
    | (L (Var a), R (Var b) ) => (Pos.eqb a b) | _ => false (* Hyp *)
    end || (
      decideOL_base n g N || ( (* Weaken *)
        match d with
        | N => decideOL_base n g g | _ => false (* Contract *)
        end || (
          match g with
          | L (Meet a b) => decideOL_base n (L a) d | _ => false (* LeftAnd1 *)
          end || (
            match g with
            | L (Meet a b) => decideOL_base n (L b) d | _ => false (* LeftAnd2 *)
            end || (
              match g with
              | L (Join a b) => decideOL_base n (L a) d && decideOL_base n (L b) d
              | _ => false (* LeftJoin *)
              end || (
                match g with
                | L (Not a) => decideOL_base n (R a) d | _ => false (* LeftNot*)
                end || (
                  ... (* Symmetric right cases *)
                  || (
                    decideOL_base n d g (* Swap *)
                    )))))))))))
  end.

```

Listing 1.4. Decision algorithm for the *word problem for ortholattices*.

```

eval ((Var 0) ∩ ¬ (Var 0)) f
  <=
eval ¬( (Var 1) ∩ ((Var 0) ∩ (Var 1))) f.

```

We can then apply a convenient lemma, itself following from the ortholattice Soundness theorem from [Listing 1.3](#) and the previously proven correctness of the algorithm `decideOLCorrect`:

```

Lemma reduce_to_decideOL {OL: Ortholattice}:
  forall t1 t2 f,
  (decideOL_base_simp (L t1) (R t2)) = true ->

```

```
((eval t1 f) <= (eval t2 f)).
```

to reduce the goal to

```
decideOL (L ((Var 0)  $\cap$   $\neg$  (Var 0)))
  (R  $\neg$ ( (Var 1)  $\cap$  ((Var 0)  $\cap$  (Var 1))))
= true.
```

Since the left hand-side is convertible to true, this is simply an instance of **reflexivity**. Effectively, we have offloaded the burden of proof to the evaluator inside Coq's kernel. This process is mechanized by the following tactic:

```
Ltac solve_OL OL :=
  reify_goal OL; apply reduceToAlgo;
  auto; vm_compute; (try reflexivity).
```

`vm_compute` is an efficient method computing the normal form of a term, in our case the evaluation of the decision procedure, while `reify_goal` computes the function `f` and the terms `s'` and `t'`, then changes the goal accordingly. To compute `f`, we put all the leaves of `s` and `t` in a list `env`. A leaf is a subexpression that is neither a meet, a join or a negation. These leaves will correspond to variables in `s'` and `t'`. `f` is then the function that maps `n` to the element in the list at place `n` (or the last element, if `n` is greater than the length of the list). The `reify_term` tactic shown next then uses this list to compute the terms `s'` and `t'`.

```
Ltac reify_term OL t env := match t with
| (?op ?X1 ?X2) => let __ := convertible op (@meet OL) in
  let r1 := reify_term OL X1 env with r2 := reify_term
    OL X2 env in
  constr:(Meet r1 r2)
| (?op ?X1 ?X2) => let __ := convertible op (@join OL) in
  let r1 := reify_term OL X1 env with r2 := reify_term
    OL X2 env in
  constr:(Join r1 r2)
| (?op ?X1) => let __ := convertible op (@neg OL) in
  let r1 := reify_term OL X1 env in
  constr:(Not r1)
| ?X1 => let j := indexOf X1 env in constr:(Var j)
end.
```

A somewhat tricky aspect of the reification is its sensibility to aliases. For example, we want our tactic to work both on

```
a && negb a = negb b && (a && b)
```

which uses the language of the specific ortholattice `bool`, but also on

```
a  $\cap$   $\neg$  a =  $\neg$  b  $\cap$  (a  $\cap$  b)
```

which uses the equivalent but different language of arbitrary ortholattices. Here, \neg is itself a notation for `@neg OL`, where `OL` is the specific ortholattice of `a` and `b`

The `convertible` tactic in `reify_term` checks if an expression `op` is convertible to `@meet OL`, `@join OL` or `@neg OL`. If so, it recursively computes the reification of the subterms. The `convertible` tactic is implemented so that it does nothing, but fails if its two arguments are not convertible:

```
Ltac convertible x y :=
  constr:(eq_refl x : x = y).
```

In the end, we obtain a tactic `solve_OL`, solving for example:

```
Lemma example: forall a b : bool,
  a && negb a = negb b && (a && b).
```

Proof.

```
intros.
solve_OL BoolOL.
```

Qed.

```
Theorem example2 {OL: Ortholattice} a b c:
  ¬(b ∪ ¬(c ∩ ¬b) ∪ a) <= ¬(a ∪ ¬(b ∩ ¬a)).
```

Proof.

```
intros.
solve_OL OL.
```

Qed.

5 Verified Memoization

The algorithm in [Listing 1.4](#) has worst case exponential runtime. Observe that since orthologic admits cut elimination, it also admits the subformula property:

Theorem 1 (Subformula Property, [11]). *If an orthologic sequent Γ, Δ has a proof, then it has a proof where only subformulas of Γ and Δ appear.*

Let n be the size of the input sequent, i.e. the number of subformulas. The number of subformulas of the initial input sequent is, and there can only exist at most $\mathcal{O}(n^2)$ different sequents built from these subformulas. Hence, in a full run of `decideOL_bool`, there can only be at most $\mathcal{O}(n^2)$ unique recursive calls. Using memoization, we can ensure that the body of the program is never executed more than $\mathcal{O}(n^2)$ times. This is essentially the same trick that makes Whitman’s algorithm quadratic instead of exponential [5].

We implement this version of the algorithm using the *state monad* paradigm. Let `MemoMap` be some type of maps with keys in `(AnTerm * AnTerm)` and values in `bool`. The function `decideOL_memo` then returns an object of type `MemoMap -> (bool, MemoMap)`. Boolean conjunctions and disjunctions are modified to compute the results in series, as in:

```

Definition mor (left : MemoMapBool ) (right : MemoMapBool) :=
  fun (memo : MemoMap) =>
    match left memo with
    | (true, m) => (true, m)
    | (false, m) => right m
    end.

```

and the decision algorithm is modified as follows:

```

Fixpoint decideOL_memo
  (fuel: nat)
  (g d: AnTerm)
  (memo: MemoMap) : (bool * MemoMap) :=
  match find (g, d) memo with
  | Some b => (b, memo)
  | None => let (b, m) := ... in
  (b, update_map (g, d) b m)
  end.

```

The correctness of the algorithm is expressed as equivalence with the non-memoized version of the algorithm, and relative to the correctness of the map given as input. A map is correct if and only if it contains only key-value pairs from the non-memoized version of the algorithm:

```

Definition memomap_correct (l: MemoMap) := forall g d,
  match find (g, d) l with
  | Some (_, true) => exists n, (decideOL_base n g d = true)
  | _ => True
  end.

```

In particular, the empty map is correct. Note that the two versions of the algorithm are not necessarily equivalent for an arbitrary allowance of fuel: The memoized version might require less. However, both algorithms are complete, and hence equivalent, for fuel greater than the size of the input sequent. The memoized algorithm's correctness is then:

```

Theorem decideOL_memo_correct :
  forall n g d l,
  (memomap_correct l) ->
  (memomap_correct (snd (decideOL_fmap n g d l))) /\
  (((fst (decideOL_fmap n g d l)) = true) -> exists n0, (decideOL_base
  n0 g d) = true).

```

The induction has to be performed jointly on the statement that the returned map is correct, and that the returned truth value is the same as that of the original algorithm.

We first implement the algorithm using lists for memoization. However, lookup inside a list takes linear time in the size of the list. As explained above,

the number of stored values is quadratic in the size of the input. Moreover, checking equality between the input and the elements of the list costs an additional linear factor. Overall, the complexity of lookup is cubic, for a total runtime of $\mathcal{O}(n^5)$.

We then implement a second version, using AVL maps from the Coq standard library. This requires to define a total order on `anTerm`, itself built on an order on `terms`, following the usual total order on labelled ordered trees. Sorted maps such as AVL maps only require a logarithmic number of comparison, down from quadratic with a list-based implementation. However, comparison still take linear time, for a total time complexity of $\mathcal{O}(n^3 \log n)$.

6 Reference Equality

In the previous two algorithms, either based on lists or on AVL maps, lookup in the map requires deciding either equality or ordering between two terms, which takes linear time in the size of the terms. However, this is *too strong* for memoization.

First, observe that if we replace this notion equality by a strictly weaker notion relation, the algorithm is still sound, and we only risk to lose the benefits of memoization.

Then note that by the subformula property, the algorithm only ever sees the $\mathcal{O}(n^2)$ different subnodes of the original input. Assigning a different binary identifier to each of these nodes requires only $\mathcal{O}(\log n)$ bits for each identifier. Then, if two terms have the same identifier, they must be structurally equal. However, the converse does not hold in general. If the original input contains multiple copies of the same subtree, they will be assigned different pointers.

This corresponds, in imperative programming, to *pointer equality*. Checking if two objects have the same location in memory is a sound approximation to deciding if they are structurally equal.

Formally, we defined an extended version of the datatype of terms:

```
Inductive TermPointer : Set :=
| VarP : nat -> Pointer -> TermPointer
| MeetP : TermPointer -> TermPointer -> Pointer -> TermPointer
| JoinP : TermPointer -> TermPointer -> Pointer -> TermPointer
| NotP : TermPointer -> Pointer -> TermPointer.
```

where `Pointer := positive` is the type of binary positive numbers. We define the projection onto regular terms `ForgetPointer : TermPointer -> Term` and a getter `GetPointer : TermPointer -> Pointer` as expected. We extends pointers to annotated pointers:

```
Inductive AnPointer : Set :=
| NP : AnPointer
| LP : Pointer -> AnPointer
| RP : Pointer -> AnPointer.
```


and `TermPointer` to `AnTermPointer` similarly. For `g` an `AnTermPointer` (that is, a term with a pointer and a left or right annotation), we note `[[g]]` its corresponding `AnPointer`. We again use AVL maps, but this time with keys being pairs of pointers. An order on pointers is provided by `Coq.PArith.BinPos`. The algorithm is modified accordingly:

```
Fixpoint decideOL_pointers
  (fuel: positive)
  (g d: AnTermPointer)
  (memo: MemoMap) : (bool * MemoMap) :=
  match M.find ([[g]], [[d]]) memo with
  | Some b => (b, memo)
  | None => let (b, m) := ... in
  (b, AnPointerPairAVLMap.add ([[g]], [[d]]) b m)
end.
```

The correctness of the algorithm of course depends on how pointers are assigned. If two different terms are assigned the same pointer, the algorithm will fail. Formally, `GetPointer` must be injective on the domain of all subterms of the input. In the correctness theorem, it is convenient to express this condition as the existence of a function `f: Pointer -> TermPointer`, corresponding to address lookup, which is the left and right inverse of `GetPointer` on all subterms of the input.

Proving the correctness extends with moderate effort from the previous algorithm with some additional side lemmas and boilerplate. However, showing that our pointer assignment is correct was much harder. We assign pointers with depth-first, preorder, traversal of the syntactic tree:

```
Fixpoint add_pointer (t: Term) (p: Pointer):
  (TermPointer * Pointer) :=
  match t with
  | Var n => (VarP n p, Pos.succ p)
  | Not a =>
    let (t1, p1) := (add_pointer a (Pos.succ p)) in (NotP t1 p, p1)
  | Meet a b =>
    let (t1, p1) := (add_pointer a (Pos.succ p)) in let (t2, p2) := (
    add_pointer b p1)
    in (MeetP t1 t2 p, p2)
  | Join a b =>
    let (t1, p1) := (add_pointer a (Pos.succ p)) in let (t2, p2) := (
    add_pointer b p1)
    in (JoinP t1 t2 p, p2)
end.
```

To construct the required inverse function, we first compute the list of subterms of a term, and map a pointer to the first term of the list with this pointer. It seems very obvious from the definition of `add_pointer` that there exists only

one such term and hence that the two functions are inverse of each other, yet this was surprisingly difficult to prove, and required a lot of intermediate lemmas about the structure of subterms, the monotonicity of `add_pointer` along subterms, correspondence between the pointer and non-pointer version of terms, and more. Overall, this part of the proof alone takes close to 800 lines in our formalization, more than the implementation and verification of the memoized version of the algorithm and as much as the definition of ortholattices and the cut elimination theorem.

7 Evaluation of Resulting Tactics

The implementations of the previous section yield four algorithms and corresponding proof tactics with different time complexity¹:

- `solve_OL` ("OL", $\mathcal{O}(2^n)$)
- `solve_OLMemoList` ("OL+l", $\mathcal{O}(n^5)$)
- `solve_OLMemoMap` ("OL+m", $\mathcal{O}(n^3 \log n)$)
- `solve_OLMemoMapPointers` ("OL+m+ϕ", $\mathcal{O}(n^2 \log n)$)

The difference of complexity is not only theoretical, but also highly observable in practice. Consider the family of propositions

$$\begin{aligned} & (x_1||x_2)\&\&(x_3||x_4)\&\&\dots\&\&(x_{n-1}||x_n) \\ & = (x_2||x_1)\&\&(x_4||x_3)\&\&\dots\&\&(x_n||x_{n-1}) \end{aligned}$$

These equalities are solvable by the laws of ortholattices. Table 2 shows the time taken by each implementation for $n = 30$. For comparison, we also include the runtime of `btauto`, an $\mathcal{O}(2^n)$ solver for boolean equalities included in Coq's standard distribution.

Tactic	Solving time [s]
<code>solve_OLMemoList</code>	1.70
<code>solve_OLMemoMap</code>	0.083
<code>solve_OLMemoMapPointers</code>	0.068
<code>btauto</code>	18.2

Table 2. Wall clock time required to prove an equality involving 30 variables.

To check instead the inequality

$$(x_1||x_2)\&\&(x_3||x_4)\&\&\dots\&\&(x_{n-1}||x_n)$$

¹ Exact complexity of algorithms up to logarithmic factor depends on the precise model of computation; for simplicity, the algorithmic complexity below assume the usual Word RAM model, where checking equality of pointers (words) takes constant time, even though this does not really correspond to Coq.




Fig. 1. Wall clock time required to prove a family of equalities with sizes ranging from 2 to 60 variables. Shaded regions indicate 95% confidence intervals. Colors indicate which implementation was used; line styles indicate which reduction strategy was used in the final step of the proof by reflection.

$$\leq (x_2||x_1)\&\&(x_4||x_3)\&\&\dots\&\&(x_n||x_{n-1}),$$

all 4 OL-based tactics take half as much time, while `btauto` essentially solves for a twice bigger formula and timeouts after 120 seconds.

To confirm the scaling characteristics of our implementation variants, we solved the family of equalities above for sizes ranging from 2 to 60 variables. Figure 1 shows our results.

All benchmarks were run on an Intel Core i9-13900K CPU with 64GB RAM.

8 Conclusion

We defined, in Coq, the algebraic structure of ortholattices. We formalized a proof system, orthologic sequent calculus, which we showed sound and complete for the class of all ortholattices. We then formalized the proof of *cut elimination for orthologic*, and discovered a missing edge case in the proof from [11].

We implemented and verified a proof search procedure for orthologic, and used reflection to obtain a proof tactic that decides equalities and inequalities in ortholattices. We improved the algorithm with important but typically imper-

ative programming features: memoization and pointer equality, improving the algorithm from exponential to quadratic.

We believe that our formalization can be useful for three different future directions. First, it opens for further development of ortholattices and orthologic, such as orthomodular lattices [14], orthologic with axioms and effectively propositional orthologic [11], interpolation in orthologic [18,10], and more [2,12,20].

Second, the OL normal form which was proposed in [8] has potential to simplify boolean formulas in Coq, but does not naturally produce a proof. Our verification of orthologic proof search and tactic solving the word problem for ortholattices allows certifying this normal form, which would instead be more efficiently implemented in Ocaml.

Finally, observe that our memoization and pointer-based optimizations, as well as the challenges resulting from their verification, are not specific to orthologic proof search. They are also of great practical interest, and necessary to obtain optimal complexity for a large range of algorithms. We expect that our formalization can be a useful source to anyone verifying efficient algorithms in the future.

References

1. Thomas Braibant, Jacques-Henri Jourdan, and David Monniaux. Implementing and Reasoning About Hash-consed Data Structures in Coq. *Journal of Automated Reasoning*, 53(3):271–304, October 2014. doi:10.1007/s10817-014-9306-0.
2. Günter Bruns. Free Ortholattices. *Canadian Journal of Mathematics*, 28(5):977–985, October 1976. doi:10.4153/CJM-1976-095-6.
3. Yannick Forster, Dominik Kirst, and Dominik Wehr. Completeness theorems for first-order logic analysed in constructive type theory: Extended version. *Journal of Logic and Computation*, 31(1):112–151, January 2021. doi:10.1093/logcom/exaa073.
4. Ralph Freese, Jaroslav Jezek, and J. Nation. *Free Lattices*, volume 42 of *Mathematical Surveys and Monographs*. American Mathematical Society, Providence, Rhode Island, March 1995. doi:10.1090/surv/042.
5. Ralph Freese and J. B. Nation. Finitely Presented Lattices. *Proceedings of the American Mathematical Society*, 77(2):174–178, 1979. arXiv:2042634, doi:10.2307/2042634.
6. Dan Frumin. Semantic Cut Elimination for the Logic of Bunched Implications, Formalized in Coq, December 2021. arXiv:2112.05515, doi:10.48550/arXiv.2112.05515.
7. G. Gentzen. Untersuchungen über das logische Schließen I. *Mathematische Zeitschrift*, 39:176–210, 1935.
8. Simon Guilloud, Mario Bucev, Dragana Milovancevic, and Viktor Kuncak. Formula Normalizations in Verification. In *35th International Conference on Computer Aided Verification*, Lecture Notes in Computer Science, pages 398–422, Paris, 2023. Springer.
9. Simon Guilloud, Sankalp Gambhir, and Viktor Kuncak. LISA – A Modern Proof System. In *14th Conference on Interactive Theorem Proving*, Leibniz International Proceedings in Informatics, pages 17:1–17:19, Bialystok, 2023. Dagstuhl.

10. Simon Guilloud, Sankalp Gambhir, and Viktor Kunčák. Interpolation and Quantifiers in Ortholattices. In *Verification, Model Checking, and Abstract Interpretation: 25th International Conference, VMCAI 2024, London, United Kingdom, January 15–16, 2024, Proceedings, Part I*, pages 235–257, Berlin, Heidelberg, January 2024. Springer-Verlag. doi:10.1007/978-3-031-50524-9_11.
11. Simon Guilloud and Viktor Kunčák. Orthologic with Axioms. *Proceedings of the ACM on Programming Languages*, 8(POPL):39:1150–39:1178, January 2024. doi:10.1145/3632881.
12. Christian Herrmann and Micheale S Roddy. On varieties of modular ortholattices which are generated by their finite-dimensional members. page 9.
13. Daniel W. H. James and R. Hinze. A Reflection-based Proof Tactic for Lattices in Coq. In *Symposium on Trends in Functional Programming*, 2009.
14. Gudrun Kalmbach. *Orthomodular Lattices*. Academic Press Inc, London ; New York, March 1983.
15. Tomoaki Kawano. Labeled Sequent Calculus for Orthologic. *Bulletin of the Section of Logic*, 47(4):217–232, December 2018. doi:10.18778/0138-0680.47.4.01.
16. Olivier Laurent. Focusing in Orthologic. In Delia Kesner and Brigitte Pientka, editors, *1st International Conference on Formal Structures for Computation and Deduction, FSCD 2016, June 22-26, 2016, Porto, Portugal*, volume 52 of *LIPICs*, pages 25:1–25:17, Porto, Portugal, 2016. Schloss Dagstuhl - Leibniz-Zentrum für Informatik. doi:10.4230/LIPICs.FSCD.2016.25.
17. Andrea Meinander. A solution of the uniform word problem for ortholattices. *Mathematical Structures in Computer Science*, 20(4):625–638, August 2010. doi:10.1017/S0960129510000125.
18. Yutaka Miyazaki. Some Properties of Orthologics. *Studia Logica: An International Journal for Symbolic Logic*, 80(1):75–93, 2005. arXiv:20016705.
19. Jürgen Schulte Mönting. Cut elimination and word problems for varieties of lattices. *Algebra Universalis*, 12(1):290–321, December 1981. doi:10.1007/BF02483891.
20. M.A.E.H. Sherif. Decision problem for orthomodular lattices. *Algebra Universalis*, 37(1):70–76, January 1997. doi:10.1007/PL00000328.
21. Matthieu Sozeau. A New Look at Generalized Rewriting in Type Theory. *Journal of Formalized Reasoning*, 2(1):41–62, 2009. doi:10.6092/issn.1972-5787/1574.
22. Hendrik Tews. Formalizing Cut Elimination of Coalgebraic Logics in Coq. In Didier Galmiche and Dominique Larchey-Wendling, editors, *Automated Reasoning with Analytic Tableaux and Related Methods*, pages 257–272, Berlin, Heidelberg, 2013. Springer. doi:10.1007/978-3-642-40537-2_22.
23. Wioletta Truskowska and Adam Grabowski. On the Two Short Axiomatizations of Ortholattices.
24. Christian Urban and Bozhi Zhu. Revisiting Cut-Elimination: One Difficult Proof Is Really a Proof. In Andrei Voronkov, editor, *Rewriting Techniques and Applications*, pages 409–424, Berlin, Heidelberg, 2008. Springer. doi:10.1007/978-3-540-70590-1_28.
25. Simon Wimmer, Shuwei Hu, and Tobias Nipkow. Verified Memoization and Dynamic Programming. In Jeremy Avigad and Assia Mahboubi, editors, *Interactive Theorem Proving*, pages 579–596, Cham, 2018. Springer International Publishing. doi:10.1007/978-3-319-94821-8_34.